
* Sascha Demerjian
* Project title: The Wages of Parenthood: At the Intersection of Sexual Orientation and Child Caregiving

/*
** HYPOTHESES

* This study has three basic stated hypotheses with a number of different dependent measures.

Stated Hyp1 - "Gay and lesbian parents receive lower assessments than their heterosexual counterparts"

Stated Hyp2 - "Mothers receive lower assessments than fathers"

Stated Hyp3 - "Primary caregivers receive lower assessments than secondary caregivers"

* Organizing the test hypotheses by outcome measure for efficiency with analyses

Suitability:

	Test-Hyp1 - Gay and lesbian parents will be seen as less suitable heterosexual parents

	Test-Hyp2 - Mothers will be seen as less suitable than fathers 

	Test-Hyp3 - Primary caregivers will be seen as less suitable than secondary caregivers 

Invite for interview:

	Test-Hyp4 - Gay and lesbian parents will rated lower on likelihood to invite to interview than heterosexual parents

	Test-Hyp5 - Mothers will rated lower on likelihood to invite to interview than fathers 

	Test-Hyp6 - Primary caregivers will rated lower on likelihood to invite to interview than secondary caregivers 

Would hire:	
	
	Test-Hyp7 - Gay and lesbian parents will rated lower on likelihood to hire than heterosexual parents

	Test-Hyp8 - Mothers will rated lower on likelihood to hire than fathers 

	Test-Hyp9 - Primary caregivers will rated lower on likelihood to hire than secondary caregivers 

Would recommend for promotion track:	
	
	Test-Hyp10 - Gay and lesbian parents will rated lower on likelihood to recommend for promotion track than heterosexual parents

	Test-Hyp11 - Mothers will rated lower on likelihood to recommend for promotion track than fathers 

	Test-Hyp12 - Primary caregivers will rated lower on likelihood to recommend for promotion track than secondary caregivers 	
	
Recommended starting salary:

	Test-Hyp13 - Gay and lesbian parents will be recommended for lower salary than heterosexual parents

	Test-Hyp14 - Mothers will be recommended for lower salary than fathers 

	Test-Hyp15 - Primary caregivers will be recommended for lower salary than secondary caregivers

Competence assessment:

	Test-Hyp16 - Gay and lesbian parents will be rated as less competent than heterosexual parents

	Test-Hyp17 - Mothers will be rated as less competent than fathers 

	Test-Hyp18 - Primary caregivers will be rated as less competent than secondary caregivers

Committment assessment:

	Test-Hyp19 - Gay and lesbian parents will be rated as less committed than heterosexual parents

	Test-Hyp20 - Mothers will be rated as less committed than fathers 

	Test-Hyp21 - Primary caregivers will be rated as less committed than secondary caregivers	

NOTES

- 	Proposal does mention that the pilot study results only supported hypotheses with some, e.g., subgrouping.
	Did not test whether these patterns in pilot data replicated but rather the hypotheses as posed.
	
-	Makes some nonspecific mention of studying interactions.  Not pursued.
	
*/

/*
import spss using TESS190_Demerjian_Client.sav, clear
save demerjian628.dta, replace
*/

clear all
use Demerjian628.dta, clear

** INDICATORS OF EXPERIMENTAL MANIPULATIONS

	recode XTESS190 (1=1) (2=1) (3=0) (4=0) (5=1) (6=1) (7=0) (8=0) (9=1) ///
		(10=1) (11=0) (12=0) (*=.), gen(vig_male)

	recode XTESS190 (1=0) (2=1) (3=0) (4=1) (5=0) (6=1) (7=0) (8=1) (9=0) ///
		(10=1) (11=0) (12=1) (*=.), gen(vig_gay)

	recode XTESS190 (1=0) (2=0) (3=0) (4=0) (5=1) (6=1) (7=1) (8=1) (9=2) ///
		(10=2) (11=2) (12=2) (*=.), gen(vig_caregiv)
	label define vig_caregive 0 "not indicated" 1 "secondary" 2 "primary"
	label values vig_caregiv vig_caregive

** OUTCOME MEASURES
	
	clonevar suitable = QW1 if QW1 >= 1 & QW1 <= 7

	clonevar invite = QW2 if QW2 >= 1 & QW2 <= 7 

	clonevar hire = QW3 if QW3 >= 1 & QW3 <= 7

	clonevar promotion = QW4 if QW4 >= 1 & QW4 <= 7

	clonevar salary = QW5 if QW5 >= 1 & QW5 <= 5

	clonevar competent = QC3 if QC3 >= 1 & QC3 <= 7

	clonevar committment = QC4 if QC4 >= 1 & QC4 <= 7

** ANALYSIS	
	
*	Test-Hyp1 - Gay and lesbian parents will be seen as less suitable heterosexual parents
regress suitable i.vig_gay
	// not supported. p=0.402
tess 1.vig_gay -, init(Demerjian628) bonf(7)
	
*	Test-Hyp2 - Mothers will be seen as less suitable than fathers 
regress suitable i.vig_male
	// not supported. p=0.225
tess 1.vig_male -, bonf(7)
	
*	Test-Hyp3 - Primary caregivers will be seen as less suitable than secondary caregivers 
regress suitable i.vig_caregiv if vig_caregiv!=0
	// not supported. p=0.684
tess 2.vig_caregiv -, bonf(7)
	
*	Test-Hyp4 - Gay and lesbian parents will rated lower on likelihood to invite to interview than heterosexual parents
regress invite i.vig_gay
	// not supported. p=0.064
tess 1.vig_gay -, bonf(7)

*	Test-Hyp5 - Mothers will rated lower on likelihood to invite to interview than fathers 
regress invite i.vig_male
	// not supported. p=0.487
tess 1.vig_male -, bonf(7)

*	Test-Hyp6 - Primary caregivers will rated lower on likelihood to invite to interview than secondary caregivers 
regress invite i.vig_caregiv if vig_caregiv!=0
	// not supported. p=0.950
tess 2.vig_caregiv -, bonf(7)
	
*	Test-Hyp7 - Gay and lesbian parents will rated lower on likelihood to hire than heterosexual parents
regress hire i.vig_gay
	// not supported. 0.330
tess 1.vig_gay -, bonf(7)
	
*	Test-Hyp8 - Mothers will rated lower on likelihood to hire than fathers 
regress hire i.vig_male
	// not supported. p=0.704
tess 1.vig_male -, bonf(7)
	
*	Test-Hyp9 - Primary caregivers will rated lower on likelihood to hire than secondary caregivers 
regress hire i.vig_caregiv if vig_caregiv!=0
	// not supported. p=0.892
tess 2.vig_caregiv -, bonf(7)
	
*	Test-Hyp10 - Gay and lesbian parents will rated lower on likelihood to recommend for promotion track than heterosexual parents
regress promotion i.vig_gay
	// not supported. p=0.153
tess 1.vig_gay -, bonf(7)

*	Test-Hyp11 - Mothers will rated lower on likelihood to recommend for promotion track than fathers 
regress promotion i.vig_male
	// not supported. p=0.754
tess 1.vig_male -, bonf(7)
	
*	Test-Hyp12 - Primary caregivers will rated lower on likelihood to recommend for promotion track than secondary caregivers 	
regress promotion i.vig_caregiv if vig_caregiv!=0	
	// not supported. p=0.781
tess 2.vig_caregiv -, bonf(7)

	
*	Test-Hyp13 - Gay and lesbian parents will be recommended for lower salary than heterosexual parents
regress salary i.vig_gay
	// not supported. p=0.892
tess 1.vig_gay -, bonf(7)
	
*	Test-Hyp14 - Mothers will be recommended for lower salary than fathers 
regress salary i.vig_male
	// not supported. p=0.734
tess 1.vig_male -, bonf(7)
	
*	Test-Hyp15 - Primary caregivers will be recommended for lower salary than secondary caregivers
regress salary i.vig_caregiv if vig_caregiv!=0
	// not supported. p=0.434
tess 2.vig_caregiv -, bonf(7)


*	Test-Hyp16 - Gay and lesbian parents will be rated as less competent than heterosexual parents
regress competent i.vig_gay
	// not supported. p=0.327
tess 1.vig_gay -, bonf(7)
	
*	Test-Hyp17 - Mothers will be rated as less competent than fathers 
regress competent i.vig_male
	// not supported. p=0.845
tess 1.vig_male -, bonf(7)
	
*	Test-Hyp18 - Primary caregivers will be rated as less competent than secondary caregivers
regress competent i.vig_caregiv if vig_caregiv!=0
	// not supported. p=0.464
tess 2.vig_caregiv -, bonf(7)


*	Test-Hyp19 - Gay and lesbian parents will be rated as less committed than heterosexual parents
regress committment i.vig_gay
	// not supported. p=0.823
tess 1.vig_gay -, bonf(7)
	
*	Test-Hyp20 - Mothers will be rated as less committed than fathers 
regress committment i.vig_male
	// not supported. p=0.697
tess 1.vig_male -, bonf(7)
	
*	Test-Hyp21 - Primary caregivers will be rated as less committed than secondary caregivers	
regress committment i.vig_caregiv if vig_caregiv!=0	
	// not supported. p=0.061	
tess 2.vig_caregiv -, bonf(7)
